// File: src/LynxVault.sol



//SPDX-License-Identifier: MIT
pragma solidity 0.8.21;

// Imports
import "openzeppelin/token/ERC20/IERC20.sol";

// Errors
error LynxVault__NotWhitelisted();
error LynxVault__NotOwner();
error LynxVault__ProperChannels();
error LynxVault__InvalidAddress();

/**
 * @title Vault for Lynx
 * @author Semi Invader
 * @notice This contract is the vault for Lynx Tokens used in other contracts.
 */
contract LynxVault {
    //------------------------
    // Variables
    //------------------------

    mapping(address => bool) public whitelisted;
    address public owner;
    IERC20 public lynx;

    //------------------------
    // Events
    //------------------------

    event SetWhitelist(address indexed _address, bool status);
    event Withdraw(address indexed _address, uint amount);

    //------------------------
    // Modifiers
    //------------------------

    modifier onlyWhitelisted() {
        if (whitelisted[msg.sender]) _;
        else revert LynxVault__NotWhitelisted();
    }

    modifier onlyOwner() {
        if (msg.sender == owner) _;
        else revert LynxVault__NotOwner();
    }

    //------------------------
    // Constructor
    //------------------------
    constructor(address _lynx) {
        whitelisted[msg.sender] = true;
        owner = msg.sender;
        lynx = IERC20(_lynx);
    }

    //----------------------------
    // External/Public functions
    //----------------------------

    /**
     * @notice Sets the whitelist status for an address
     * @param _address Address to set the whitelist status for
     * @param status the status to set the whitelist to
     */
    function setWhitelistStatus(
        address _address,
        bool status
    ) external onlyOwner {
        if (_address == address(0)) revert LynxVault__InvalidAddress();
        whitelisted[_address] = status;
        emit SetWhitelist(_address, status);
    }

    /**
     * @notice Sets the whitelist status for multiple Addresses
     * @param _addresses Addresses to set the whitelist status for
     * @param status the status value set
     */
    function addMultipleWhitelist(
        address[] calldata _addresses,
        bool status
    ) external onlyOwner {
        for (uint i = 0; i < _addresses.length; i++) {
            if (_addresses[i] == address(0)) revert LynxVault__InvalidAddress();
            whitelisted[_addresses[i]] = status;
            emit SetWhitelist(_addresses[i], status);
        }
    }

    function withdraw(uint amount) external onlyWhitelisted {
        lynx.transfer(msg.sender, amount);
        emit Withdraw(msg.sender, amount);
    }

    function withdrawTo(
        address _address,
        uint amount
    ) external onlyWhitelisted {
        if (_address == address(0)) revert LynxVault__InvalidAddress();
        lynx.transfer(_address, amount);
        emit Withdraw(_address, amount);
    }

    function recoverERC20(address _otherToken) external onlyOwner {
        if (_otherToken == address(lynx)) revert LynxVault__ProperChannels();
        IERC20 otherToken = IERC20(_otherToken);
        otherToken.transfer(msg.sender, otherToken.balanceOf(address(this)));
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}